```
; Subroutine FASTCLS
; Purpose - to  clear the
; screen as fast as possible.

   ORG   &9000       ;                                  T-states
   DI                ; DI for nigher speed                 4
   PUSH    HL        ; Store all registers used           11
   PUSH    DE                                             11
   PUSH    BC                                             11
   PUSH    AF                                             11
   LD    C,&08       ; 8 lines to clear                    7
   LD    (STAK),SP   ; save SP                            20
   LD    DE,&0080    ; offset between blocks               10
   LD    SP,&FF80    ; top of screen                       10
OLP
   LD    H,D         ; bytes to put into screen            4)
   LD    L,D         ; quicker than LD HL,0                4)
   LD    B,&3C       ; counter 60 times                    7)
ILP
   PUSH HL           ; clear 2 bytes 16 times
   PUSH  HL
   PUSH  HL          ; NB
   PUSH  HL          ; 11*16=176 Tstates
   PUSH  HL          ; for these pushes
   PUSH  HL          ; add 13 for DJNZ
   PUSH  HL          ; therefore 189 T-states
   PUSH  HL          ; for this inner loop
   PUSH  HL          ; except last when 184
   PUSH  HL          ; 60*189=11340 - 5
   PUSH  HL          ; total =11335 T states for
   PUSH  HL          ; everything in ILP
   PUSH  HL
   PUSH  HL
   PUSH  HL
```

```
   PUSH  HL                                        11335)
   DJNZ  ILP        ; repeat
   ADD   HL,SP      ; SP into HL                      11)
   AND   A          ; clear Carry                      4)
   SBC   HL,DE      ; subtract offset                 15)
   LD    SP,HL      ; into stack pointer again        11)
   DEC   C          ; decrement block counter          4)
   JR    NZ,OLP     ; loop back for next block        12)
                    ;                              -5 for true
                    ;                          11407*8=91256
;  less 5 once when JR is false    92272-5=          91251
   LD    SP,(STAK) ; restore stack pointer              20
   POP     AF      ; Restore registers                  11
   POP     BC                                           11
   POP     DE                                           11
   POP     HL                                           11
   EI               ; restore interrupts                  4
   RET              ; and return                         10
STAK
   DEFW  0          ; to store stack pointer

                    ;                 Total=91444 T-states
                    ;                 at 4 MHz=.022 secs

;  Compare with
;  LD    HL,&C000    10
;  LD    DE,&C001    10
;  LD    BC,&3FFF    10
;  LD    (HL),0      10
;  LDIR              21*&3FFE+16 when BC=0 =344038
;  RET               10

;  Total 344088 T states at 4 Mhz=.08 secs.
```